Pensieve header: Notebook for Dalyell meeting 1: the Jones polynomial.

Find everything at http://drorbn.net/syd3

```
Setting Knots to contain all from Knot[3,1] to Knot[10,165] as well as GST48...
```

Out[•]=



In[•]:=
```
1 + 1
```
Out[•]= 2

In[•]:=
```
Integrate[1 / (1 + x^2), x]
```
Out[•]= ArcTan[x]

In[•]:=
```
K = PD[X[1, 5, 2, 4], X[5, 3, 6, 2], X[3, 1, 4, 6]]
```
Out[•]= PD[X[1, 5, 2, 4], X[5, 3, 6, 2], X[3, 1, 4, 6]]

In[•]:=
```
prod = 1;
Do[prod = prod * k, {k, 1, 7}];
prod
```
Out[•]= 5040

In[•]:=
```
{10, 9, 180, 26} /. 9 → 23
```
Out[•]= {10, 23, 180, 26}

In[•]:=
```
{10, 9, 180, 26} /. x_ → x^2
```
Out[•]= {100, 81, 32400, 676}

In[•]:=
```
{10, 9, 180, 26, Zsuzsi} /. x_ → x^2
```
Out[•]= $\{100, 81, 32400, 676, \text{Zsuzsi}^2\}$

```
In[•]:= t1 = K /. X[i_, j_, k_, l_] → A * p[i, j] * p[k, l] + B * p[i, l] p[j, k]
Out[•]= PD[A p[1, 5] p[2, 4] + B p[1, 4] p[5, 2],
          B p[3, 6] p[5, 2] + A p[5, 3] p[6, 2], B p[1, 4] p[3, 6] + A p[3, 1] p[4, 6]]

In[•]:= t1
Out[•]= PD[A p[1, 5] p[2, 4] + B p[1, 4] p[5, 2],
          B p[3, 6] p[5, 2] + A p[5, 3] p[6, 2], B p[1, 4] p[3, 6] + A p[3, 1] p[4, 6]]

In[•]:= Expand[(a + b) (a - b)]
Out[•]= a^2 - b^2

In[•]:= Factor[a^2 - b^2]
Out[•]= (a - b) (a + b)

In[•]:= Expand[t1]
Out[•]= PD[A p[1, 5] p[2, 4] + B p[1, 4] p[5, 2],
          B p[3, 6] p[5, 2] + A p[5, 3] p[6, 2], B p[1, 4] p[3, 6] + A p[3, 1] p[4, 6]]

In[•]:= Factor[a^2 - b^2] // FullForm
Out[•]//FullForm= Times[Plus[a, Times[-1, b]], Plus[a, b]]

In[•]:= t2 = Expand[t1 /. PD → Times]
Out[•]= A B^2 p[1, 4] p[1, 5] p[2, 4] p[3, 6]^2 p[5, 2] + A^2 B p[1, 5] p[2, 4] p[3, 1] p[3, 6] p[4, 6] p[5, 2] +
          B^3 p[1, 4]^2 p[3, 6]^2 p[5, 2]^2 + A B^2 p[1, 4] p[3, 1] p[3, 6] p[4, 6] p[5, 2]^2 +
          A^2 B p[1, 4] p[1, 5] p[2, 4] p[3, 6] p[5, 3] p[6, 2] +
          A^3 p[1, 5] p[2, 4] p[3, 1] p[4, 6] p[5, 3] p[6, 2] +
          A B^2 p[1, 4]^2 p[3, 6] p[5, 2] p[5, 3] p[6, 2] + A^2 B p[1, 4] p[3, 1] p[4, 6] p[5, 2] p[5, 3] p[6, 2]

In[•]:= {7, 1} /. {{0, n_} → n, {k_, n_} → {k - 1, n * k}}
Out[•]= {6, 7}

In[•]:= {7, 1} //. {{0, n_} → n, {k_, n_} → {k - 1, n * k}}
Out[•]= 5040

In[•]:= t3 = t2 //. p[i_, j_] p[j_, k_] → p[i, k]
Out[•]= A^2 B p[1, 4]^2 + A^3 p[2, 2] p[3, 3] + A^2 B p[3, 6]^2 + A B^2 p[1, 4]^2 p[3, 6]^2 + A^2 B p[5, 2]^2 +
          A B^2 p[1, 4]^2 p[5, 2]^2 + A B^2 p[3, 6]^2 p[5, 2]^2 + B^3 p[1, 4]^2 p[3, 6]^2 p[5, 2]^2

In[•]:= t4 = t3 /. {p[i_, i_] → d, p[i_, j_]^2 → d}
Out[•]= 3 A^2 B d + A^3 d^2 + 3 A B^2 d^2 + B^3 d^3

In[•]:= t5 = Expand[t4 /. {B → 1 / A, d → -A^2 - 1 / A^2}]
Out[•]= -1/A^9 + 1/A + A^3 + A^7
```

In[•]:=
```
Knots
```

Out[•]=
```
{Knot[3, 1] → PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]], ··· 248 ···,
 GST48 → PD[X[1, 15, 2, 14], X[29, 2, 30, 3], X[40, 4, 41, 3], X[4, 44, 5, 43], ··· 40 ···,
   X[94, 67, 95, 68], X[86, 75, 87, 76], X[77, 88, 78, 89], X[93, 78, 94, 79]]}
```

large output | show less | show more | show all | set size limit...

In[•]:=
```
Knot[3, 1] /. Knots
```

Out[•]=
```
PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]]
```

In[•]:=
```
Knot[10, 165] /. Knots
```

Out[•]=
```
PD[X[1, 6, 2, 7], X[7, 18, 8, 19], X[3, 9, 4, 8], X[17, 3, 18, 2], X[5, 15, 6, 14],
 X[9, 17, 10, 16], X[15, 11, 16, 10], X[11, 5, 12, 4], X[20, 14, 1, 13], X[12, 20, 13, 19]]
```

In[•]:=
```
K = Knot[10, 73] /. Knots;
t1 = K /. X[i_, j_, k_, l_] → A * p[i, j] * p[k, l] + B * p[i, l] p[j, k];
t2 = Expand[t1 /. PD → Times];
t3 = t2 //. {p[i_, j_] p[j_, k_] → p[i, k], p[i_, j_] p[k_, j_] → p[i, k]};
t4 = t3 /. {p[i_, i_] → d, p[i_, j_]^2 → d};
Expand[t4 /. {B → 1 / A, d → -A^2 - 1 / A^2}]
```

Out[•]=
$$\frac{1}{A^{22}} - \frac{3}{A^{18}} + \frac{3}{A^{14}} - \frac{4}{A^{10}} + \frac{2}{A^6} - \frac{1}{A^2} - A^2 + 3A^6 - 4A^{10} + 3A^{14} - 2A^{18} + A^{22}$$

In[•]:=
```
K = GST48 /. Knots
```

Out[•]=
```
PD[X[1, 15, 2, 14], X[29, 2, 30, 3], X[40, 4, 41, 3], X[4, 44, 5, 43], X[5, 26, 6, 27],
 X[95, 7, 96, 6], X[7, 1, 8, 96], X[8, 14, 9, 13], X[28, 9, 29, 10], X[41, 11, 42, 10],
 X[11, 43, 12, 42], X[12, 27, 13, 28], X[15, 31, 16, 30], X[61, 16, 62, 17], X[72, 17, 73, 18],
 X[83, 18, 84, 19], X[34, 20, 35, 19], X[20, 89, 21, 90], X[92, 21, 93, 22], X[22, 79, 23, 80],
 X[23, 68, 24, 69], X[24, 57, 25, 58], X[56, 25, 57, 26], X[31, 63, 32, 62], X[32, 74, 33, 73],
 X[33, 85, 34, 84], X[35, 50, 36, 51], X[81, 37, 82, 36], X[70, 38, 71, 37], X[59, 39, 60, 38],
 X[54, 39, 55, 40], X[55, 45, 56, 44], X[45, 59, 46, 58], X[46, 70, 47, 69],
 X[47, 81, 48, 80], X[91, 49, 92, 48], X[49, 91, 50, 90], X[82, 52, 83, 51], X[71, 53, 72, 52],
 X[60, 54, 61, 53], X[74, 63, 75, 64], X[85, 64, 86, 65], X[65, 76, 66, 77],
 X[66, 87, 67, 88], X[94, 67, 95, 68], X[86, 75, 87, 76], X[77, 88, 78, 89], X[93, 78, 94, 79]]
```

In[•]:=
```
Length[K]
```

Out[•]=
```
48
```

```
In[•]:= K = GST48 /. Knots;
        t1 = K /. X[i_, j_, k_, l_] → A * p[i, j] * p[k, l] + B * p[i, l] p[j, k];
        t2 = Expand[t1 /. PD → Times];
        t3 = t2 //. {p[i_, j_] p[j_, k_] → p[i, k],
            p[i_, j_] p[k_, j_] → p[i, k], p[j_, i_] p[j_, k_] → p[i, k]};
        t4 = t3 /. {p[i_, i_] → d, p[i_, j_]^2 → d};
        Expand[t4 /. {B → 1 / A, d → -A^2 - 1 / A^2}]
Out[•]= $Aborted

In[•]:= K = Knot[8, 17] /. Knots;
        SetAttributes[p, Orderless];
        t1 = K /. X[i_, j_, k_, l_] → A * p[i, j] * p[k, l] + B * p[i, l] p[j, k];
        t2 = Expand[t1 /. PD → Times];
        t3 = t2 //. {p[i_, j_] p[j_, k_] → p[i, k]};
        t4 = t3 /. {p[i_, i_] → d, p[i_, j_]^2 → d};
        Expand[t4 /. {B → 1 / A, d → -A^2 - 1 / A^2}]
```

$$-\frac{1}{A^{18}}+\frac{2}{A^{14}}-\frac{2}{A^{10}}+\frac{1}{A^{6}}-\frac{1}{A^{2}}-A^{2}+A^{6}-2\,A^{10}+2\,A^{14}-A^{18}$$

Our blackboard: